# Maintainer: Levente Polyak <anthraxx[at]archlinux[dot]org>
# Contributor: brent s. <bts[at]square-r00t[dot]net>
# Contributor: Joris Steyn <jorissteyn@gmail.com>
# Contributor: Ondrej Kucera <ondrej.kucera@centrum.cz>

pkgname=httping
pkgver=2.9
pkgrel=1
pkgdesc='A ping-like tool for http-requests'
url='https://www.vanheusden.com/httping/'
arch=('loong64' 'x86_64')
license=('GPL3')
depends=('openssl' 'ncurses' 'fftw')
source=("${pkgname}-${pkgver}.tar.gz::https://github.com/folkertvanheusden/HTTPing/archive/refs/tags/v${pkgver}.tar.gz")
sha512sums=('74d4a98590b6862a2f15fffdf3bc58e3226a94c7aebbe60135695397b20e6cb71fe8c2af75b8525724f2f9ce6530d1f8eecd3193bc50f31d254bbb9903615019')

build() {
  cd HTTPing-${pkgver}
  ./configure --with-tfo
  make
}

package() {
  cd HTTPing-${pkgver}
  make DESTDIR="${pkgdir}" install
}

# vim: ts=2 sw=2 et:
